Reading STEREO ephemerides as SPICE kernels

William Thompson May 13, 2004

1 Introduction

The STEREO orbit and attitude data will be distributed as SPICE kernel files. Many members of the STEREO team may not be familiar with the SPICE library. Therefore, this document will serve as an introduction to the SPICE system, and how it applies to STEREO.

As of this initial writing, only a sample of the STEREO ephemeris is available. Therefore, for the moment, we will only address the question of extracting the positions of the spacecraft. We will extend this introduction to include orientation when sample attitude files are also available.

SPICE is available as both FORTRAN and C libraries. The CSPICE library can also be used from IDL as a Dynamically Loadable Module (DLM). We will concentrate on the IDL interface, called Icy, but the discussion is also applicable to the FORTRAN and C libraries with little modification.

2 Installing SPICE

The SPICE software and associated data files are available via ftp from

```
ftp://naif.jpl.nasa.gov/pub/naif/
```

The software is found under the directory toolkit, and then further divided by software environment and operating system. For example, the compressed tar file

```
ftp://naif.jpl.nasa.gov/pub/naif/toolkit/IDL/PC_Linux_C/packages/icy.tar.Z
```

contains everything needed to install and run Icy/CSPICE on Linux. The files can be unpacked into a suitable location, e.g. /usr/local/icy. No compilation should be needed, but the file makeall.csh can be used to recompile the software if necessary.

The Icy/CSPICE tar package will include the following directories

data	Sample data files used with the "cookbook" routines
doc	Documentation, in both plain-text and HTML formats
exe	Executable utility files
lib	Object libraries and idl.dlm
<pre>src/icycook</pre>	Sample IDL "cookbook" procedures

plus other directories not discussed here. It's instructive to look over the IDL "cookbook" procedures as an introduction to using Icy/CSPICE.

SPICE consists of two parts: the library itself, and the a collection of stand-alone utilities in the exe directory. The latter are mainly used for managing the contents of kernel files. For example, the toxfer program converts binary kernel files into an ASCII transfer format, while tobin does the reverse.

In order to use the Icy/CSPICE library, the environment variable IDL_DLM_PATH must be defined before starting IDL, to point to the directory containing the shareable object library and DLM file, e.g.

setenv IDL_DLM_PATH "/usr/local/rsi/idl/bin/bin.linux.x86:/usr/local/icy/lib"

It's important to include the standard path in the definition of IDL_DLM_PATH. To determine what this should be, one should print out !DLM_PATH in IDL before defining the environment variable.

In IDL, all SPICE routines start with the prefice "CSPICE_", for example CSPICE_SPKEZR, CSPICE_SUBPT, etc.

3 SPICE kernels

The term "kernels" is used to denote data files used by the SPICE system. At the time of writing, the sample STEREO orbit kernel is stereo_20060211a_nominal.xsp with a creation date of February 6, 2004. This is an example of what is called an SPK, or Spacecraft and Planet Kernel. The file extension .xsp means that it's in a platform-independent ASCII transfer format. This needs to be converted into a binary file before it can be used. The tobin utility in the exe directory will convert the ASCII file into a binary file with the extension .bsp. These binary files are platform-specific, but modern versions of SPICE can read binary files from other platforms, with the exception of OpenVMS.

The brief program can be used to get a brief summary of the contents of a binary SPK file. The results for the sample STEREO ephemeris file are shown in Figure 1. This tells us that the file contains ephemerides for two bodies, named "STEREO AHEAD" and "STEREO BEHIND", with ID numbers -234 and -235 respectively. (Spacecraft ID numbers are always negative in SPICE, while planets and other solar system bodies have positive ID numbers.) Also, we have the date and time ranges that the file is valid for. (SPICE uses ephemeris time (ET) which has a small offset from TAI or UTC—this will be discussed in more detail later.) The spacit utility can be used to display more detailed information about the file.

Although the file stereo_20060211a_nominal.bsp contains all the ephemeris data needed for the two STEREO spacecraft, it cannot be used by itself; it needs to be put into its context within the solar system. Specifically, the positions of the two STEREO are defined relative to Earth early in the mission, and relative to the Sun later in the mission, while SPICE internally needs to relate all of these to the solar system barycenter. Thus, a solar system ephemeris needs to be loaded. There are several generic solar system kernels available, but the correct one to use is de405.bsp, because that was the kernel that was used to generate the STEREO ephemeris.

Another data file required by SPICE is a Leapseconds Kernel (LSK), to support various time conversions. The most recent leapseconds file is naif0007.tls.

A Planetary Constants Kernel (PCK) allows coordinates to be expressed in longitude and latitude. For example, pck00007.tpc is a fairly modern and comprehensive PCK for the solar

Figure 1: Synopsis of the contents of the sample STEREO ephemeris file.

system as a whole. Using this file allows one to extract either geographic or heliographic (i.e. Carrington) coordinates.

All of these files can be downloaded from

```
ftp://naif.jpl.nasa.gov/pub/naif/generic_kernels
```

4 Basic software calls

4.1 Loading SPICE kernels

SPICE kernels are loaded with calls to CSPICE_FURNSH, one for each kernel file. For example, to load the STEREO ephemeris, together with the generic planet ephemeris and leapseconds, one would type

```
cspice_furnsh, 'naif0007.tls'
cspice_furnsh, 'de405.bsp'
cspice_furnsh, 'stereo_20060211a_nominal.bsp'
```

Assuming that each of these files are in the current directory. Similarly, calls to CSPICE_UNLOAD can be used to unload kernels once they are no longer needed.

4.2 Time

Internally, SPICE uses Ephemeris Time (ET), also known as Barycentric Dynamical Time (TDB). This time varies from TAI time by 32.184 seconds plus a small annual variation on the order of a few milliseconds. In SPICE, ET is expressed as the number of dynamical seconds since 1-Jan-2000 12:00 ET.

The routine CSPICE_STR2ET converts a wide variety of time string formats into double precision ET values. A simple way to convert time values into ET format is to first use the SolarSoft routine ANYTIM2UTC with the /CCSDS keyword to convert date/time values into ISO 8601 format, which SPICE is able to ingest, e.g.

Table 1: NAIF ID names and numbers for various solar system objects, and the two STEREO spacecraft. Character string names are case-insensitive.

MERCURY BARYCENTER	1	SATURN BARYCENTER	6	MERCURY	199
VENUS BARYCENTER	2	URANUS BARYCENTER	7	VENUS	299
EARTH BARYCENTER	3	NEPTUNE BARYCENTER	8	MOON	301
MARS BARYCENTER	4	PLUTO BARYCENTER	9	EARTH	399
JUPITER BARYCENTER	5	SUN	10	MARS	499
STEREO AHEAD -234		SOLAR SYSTEM BARYCENTER (or SSB)			0
STEREO BEHIND -235		EARTH MOON BARYCENTER (or EMB)			3

```
utc = anytim2utc(date_time, /ccsds)
cspice_str2et, utc, et
```

(The CCSDS keyword refers to the Consultative Committee for Space Data Systems, which endorsed a specific subset of the wider ISO-8601 standard.)

Similarly, the routine CSPICE_ET2UTC can be used to convert from ephemeris time into a UTC character string. For example, the call

```
cspice_et2utc, et, 'ISOC', 3, utcstr
```

would return a string such as "2000-01-01T12:00:00.000". The "ISOC" tells the software to use ISO 8601 calendar format, and the "3" tells the software to return seconds to three significant figures past the decimal point.

4.3 Spacecraft and planet identifiers

Spacecraft and planets are identified within SPICE in one of two ways: either by a character string name, or by an ID number. Table 1 shows the ID numbers for various solar system objects, and the two STEREO spacecraft. By convention, spacecraft IDs are always negative.

Internally, SPICE uses ID numbers to track spacecraft and solar system objects, and lower level routines take the ID numbers as input parameters. Higher level routines, however, use character string names, such as "Earth" or "Sun". A few high level routines can take either the name or the ID number, provided that the ID number is formatted as a character string.

4.4 Reference frames

SPICE incorporates two basic kinds of reference frames: inertial and body-fixed frames. Examples of inertial frames include:

J2000: Celestial reference frame, with Z aligned with Earth's rotational axis, and X aligned with the direction of the vernal equinox, at the dynamical epoch of J2000.

ECLIPJ2000: Ecliptic coordinates based upon the J2000 frame.

among others. Note that these inertial frames are based on the Earth's rotational axis and ecliptic for a specific date, and do not take precession into account. This will be discussed in more detail later.

Body-fixed frames include geographic and Carrington heliographic, as well as spacecraft and instrument-based frames. The reference frames for geographic and Carrington heliographic coordinates are IAU_EARTH and IAU_SUN respectively.

The routine CSPICE_SPKEZR provides a simple interface to extract inertial frame coordinates of objects. The calling sequence is

```
CSPICE_SPKEZR, TARGET, ET, REF, ABCORR, OBS, STATE, LTIME
```

where the input parameters are

TARGET: Target body name.

ET: Ephemeris time of the observation.

REF: Reference frame, e.g. 'ECLIPJ2000' or 'IAU_SUN'.

ABCORR: Aberration correction. Possible values include:

'NONE' No aberration correction

'LT' Correct for one-way light travel time

'LT+S' Correct for one-way light travel time, and stellar aberration

One can also use 'XLT' and 'XLT+S' in the transmission case, where ET then refers to the time of transmission, and ET+LTIME is the observation time.

OBS: The name of the observing body; i.e. the origin of the coordinate system.

and the output parameters are

STATE: The six-element state vector of the target. The first three parameters are the position in kilometers, and the last three are the velocity in km/sec.

LTIME: The one-way light travel time from the target to the observer.

The target and observer parameters can be either their character string names, or their ID numbers formatted as a character string. For example, the call

```
cspice_spkezr, '-234', et, 'J2000', 'LT+S', 'Earth', state, ltime
```

returns the state vector of the STEREO Ahead spacecraft as seen from Earth in the J2000 celestial coordinate system. The complete process from loading to unloading the kernels is presented in Figure 2.

```
pro stereo_a_demo, date, state, ltime
;
cspice_furnsh, 'naif0007.tls'
cspice_furnsh, 'de405.bsp'
cspice_furnsh, 'pck00007.tpc'
cspice_furnsh, 'stereo_20060211a_nominal.bsp'
;
utc = anytim2utc(date,/ccsds)
cspice_str2et, utc, et
;
cspice_spkezr, 'STEREO AHEAD', et, 'IAU_SUN', 'None', 'Sun', state, ltime
;
cspice_unload, 'stereo_20060211a_nominal.bsp'
cspice_unload, 'pck00007.tpc'
cspice_unload, 'de405.bsp'
cspice_unload, 'naif0007.tls'
;
end
```

Figure 2: Sample IDL program for deriving the position and velocity of the STEREO A spacecraft in Carrington heliographic coordinates at a given date.

5 Coordinate systems

It is possible to use the SPICE software to derive coordinates in a number of standard coordinate systems, listed below. Two axes are chosen to define each coordinate system, and the missing axis satisfies the right-hand rule. For simplicity, all coordinate systems are described in the J2000 reference frame, without corrections for precession.

GEI: Geocentric Equatorial Inertial. The X axis points toward the first point of Aries (i.e. the vernal equinox), and the Z axis is aligned with the geographic north pole. When expressed in terms of longitude and latitude, this is the well known celestial coordinate system of right ascension and declination.

```
cspice_spkezr, target, et, 'J2000', abcorr, 'Earth', state, ltime
```

GEO: Geographic. X is the intersection of the Greenwich meridian and the geographic equator, and Z is the geographic North Pole. This is usually expressed as geographic longitude and latitude.

```
cspice_spkezr, target, et, 'IAU_EARTH', abcorr, 'Earth', state, ltime
```

(However, see Section 7 for a discussion of the distinction between planetographic and planetocentric coordinates.)

GSE: Geocentric Solar Ecliptic. X is the Earth-Sun line, and Z is the ecliptic North Pole. This is calculated by calculating the ecliptic coordinates of both the target and the Sun, and using the latter to define a rotation about the Z axis.

```
cspice_spkezr, target, et, 'ECLIPJ2000', abcorr, 'Earth', state, ltime
cspice_spkezr, 'Sun', et, 'ECLIPJ2000', abcorr, 'Earth', ssun, ltsun
rho = sqrt(state[0]^2 + state[1]^2)
vrho = sqrt(state[3]^2 + state[4]^2)
theta_sun = atan(ssun[1], ssun[0])
theta = atan(state[1], state[0]) - theta_sun
vtheta = atan(state[4], state[3]) - theta_sun
state[0] = rho*cos(theta)
state[1] = rho*sin(theta)
state[3] = vrho*cos(vtheta)
state[4] = vrho*sin(vtheta)
```

HAE: Heliocentric Aries Ecliptic. X is in the direction of the first point of Aries, and Z is the ecliptic North Pole.

```
cspice_spkezr, target, et, 'ECLIPJ2000', abcorr, 'Sun', state, ltime
```

HEE: Heliocentric Earth Ecliptic. X is the Sun-Earth line, and Z is the Ecliptic North Pole.

```
cspice_spkezr, target, et, 'ECLIPJ2000', abcorr, 'Sun', state, ltime
cspice_spkezr, 'Earth', et, 'ECLIPJ2000', abcorr, 'Sun', searth, ltearth
rho = sqrt(state[0]^2 + state[1]^2)
vrho = sqrt(state[3]^2 + state[4]^2)
theta_earth = atan(searth[1], searth[0])
theta = atan(state[1], state[0]) - theta_earth
vtheta = atan(state[4], state[3]) - theta_earth
state[0] = rho*cos(theta)
state[1] = rho*sin(theta)
state[3] = vrho*cos(vtheta)
state[4] = vrho*sin(vtheta)
```

HEEQ: Heliocentric Earth Equatorial. Z is the solar rotation axis, and X is in the plane containing the Z axis and Earth, at the intersection of the solar central meridian, and the heliographic equator. When converted to longitude and latitude, this is known as Stonyhurst heliographic coordinates.

```
cspice_spkezr, target, et, 'IAU_SUN', abcorr, 'Sun', state, ltime
cspice_spkezr, 'Earth', et, 'IAU_SUN', abcorr, 'Sun', searth, ltearth
rho = sqrt(state[0]^2 + state[1]^2)
vrho = sqrt(state[3]^2 + state[4]^2)
theta_earth = atan(searth[1], searth[0])
theta = atan(state[1], state[0]) - theta_earth
vtheta = atan(state[4], state[3]) - theta_earth
state[0] = rho*cos(theta)
state[1] = rho*sin(theta)
state[3] = vrho*cos(vtheta)
state[4] = vrho*sin(vtheta)
```

Carrington heliographic. Similar to HEEQ in that Z is the solar rotational axis. The X axis is at the intersection of the Carrington prime meridian and the heliographic equator. The

coordinate system rotates with a sidereal period of 25.38 days. This coordinate system is usually expressed in terms of longitude and latitude.

```
cspice_spkezr, target, et, 'IAU_SUN', abcorr, 'Sun', state, ltime
```

Any of the above coordinate systems can be converted into longitude and latitude by using the CSPICE calls

```
cspice_reclat, state[0:2], radius, lon, lat
lon = lon * cspice_dpr()
lat = lat * cspice_dpr()

or by using the IDL calls

radius = sqrt(total(state[0:2]^2))
lon = (180/!dpi) * atan(state[1], state[0])
lat = (180/!dpi) * atan(state[2], sqrt(total(state[0:2]^2)))
```

6 Precession

So far, all of the coordinate systems have been discussed in terms of the J2000 inertial reference frame. However, Earth's rotational axis slowly precesses around the pole of the ecliptic at a rate of about 50" per year with a period of about 26,000 years. Also, the plane of the ecliptic itself precesses at a rate of about 47" per century. There may be situations where these effects need to be taken into account.

Fränz and Harper (2002), discuss the effects of precession on the standard heliospheric coordinate systems, and distinguish between three different versions of the Geocentric Equatorial Inertial system:

```
GEI<sub>J2000</sub> Earth mean equator at J2000.

GEI<sub>D</sub> Earth mean equator of date, i.e. corrected for precession.

GEI<sub>T</sub> Earth true equator of date, i.e. corrected for precession and nutation.
```

SPICE returns GEI_{J2000} . On the other hand, geographic coordinates (GEO) are defined relative to the true equator of date, which is how they are implemented in SPICE.

Similarly, Franz and Harper distinguish between two versions of Heliocentric Aries Ecliptic:

```
HAE_{J2000} Earth mean ecliptic at J2000. 
 HAE_{D} Earth mean ecliptic of date.
```

Again, SPICE returns HAE_{J2000} . However, the other ecliptic systems, GSE and HEE, are explicitly defined by Fränz and Harper in terms of the mean ecliptic of date, while SPICE wants to work within the J2000 reference frame. To solve this problem, the routine ECLIPJ2000T0DATE is used to convert ecliptic coordinates from the J2000 reference frame to the mean ecliptic of date, and vice versa. For example, to retrieve HAE_D coordinates, one would use

```
cspice_spkezr, target, et, 'ECLIPJ2000', abcorr, 'Sun', state, ltime eclipj2000todate, et, state
```

For GSE and HEE coordinates, this precession is applied to both state vectors (state and either ssun or searth in Section 5) just after the CSPICE_SPKEZR calls, before the coordinate rotation is calculated. ECLIPJ2000TODATE should not be used for non-ecliptic coordinate systems.

Precession is not a concern for HEEQ or Carrington heliographic coordinates.

7 Planetographic versus planetocentric coordinates

In cartography, a distinction is made between planetocentric and planetographic coordinates. Planetocentric latitude is the angle between the equatorial plane and a line drawn from the center of mass to the reference point. Planetographic latitude, however, is defined in terms of a line which is normal to a spheroidal reference surface for the body in question. For spherical bodies, such as the Sun, there is no distinction between these two kinds of latitudes, but there is a difference for an oblate spheroid such as Earth. For the STEREO spacecraft, the distinction between geocentric and geographic latitude varies between about $\pm 0^{\circ}.15$. The coordinates returned by the CSPICE_SPKEZR call in Section 5 are in the planetocentric system.

There's also a distinction between planetocentric and planetographic longitude when the body's rotation is retrograde, but this does not apply to either Earth or the Sun.

Fränz and Harper (2002) explicitly define the GEO coordinate system as a planetographic system. However, Russell (1971) and Hapgood (1992), ignore the distinction between planetographic and planetocentric in their discussions of the GEO coordinate system and its relationship to the other coordinate systems used here. It's not clear to this author which convention is more commonly used for magnetospheric and heliospheric data.

Only planetocentric coordinates can be expressed as a true X,Y,Z coordinate system. When planetographic latitude and longitude are converted into distances using the standard trigonometric transformations, the metric along the Z axis varies with position. Thus, it is recommended that planetographic coordinates be restricted to latitude and longitude, while X,Y,Z coordinates are always planetocentric.

Although SPICE returns planetocentric coordinates, the routines CSPICE_BODVAR and CSPICE_RECGEO can be used to derive planetographic longitude and latitude, as well as the altitude above the surface, e.g.

```
cspice_spkezr, 'STEREO Ahead', et, 'IAU_EARTH', abcorr, 'Earth', state, ltime
cspice_bodvar, 399, 'RADII', radii
flat = (radii[0]-radii[2]) / radii[0]
cspice_recgeo, state[0:2], radii[0], flat, lon, lat, alt
lon = lon * (180/!dpi)
lat = lat * (180/!dpi)
```

Note that CSPICE_BODVAR requires the body ID number, and that 'RADII' must be uppercase.

8 Using CSPICE_SUBPT

An alternative way to derive heliographic coordinates is to use CSPICE_SUBPT, which returns the sub-observer point on a spheroidal body. The advantage of using it over CSPICE_SPKEZR is that CSPICE_SUBPT treats the STEREO spacecraft as an observer rather than as a target. The calling sequence is

```
CSPICE_SUBPT, METHOD, TARGET, ET, ABCORR, OBS, SPOINT, ALT
```

where the input parameters are

METHOD: Calculation method. Possible values are:

'NEAR POINT' Nearest point on surface to observer

'INTERCEPT' Intercept point along line from target center

For a spherical object like the Sun, the two methods return the same result. The 'INTERCEPT' method is less computationally intensive.

TARGET: Target body name, e.g. 'Sun'.

ET: Ephemeris time of the observation.

ABCORR: Aberration correction. (See Section 4.4.)

OBS: The name of the observing body, e.g. 'STEREO AHEAD'.

and the output parameters are

SPOINT: Planetocentric coordinates of the sub-observer point on the target surface.

ALT: The altitude, calculated as the distance from the sub-observer point to the observer.

The following example shows the derivation of Carrington heliographic longitude and latitude using CSPICE_SUBPT

```
cspice_subpt, 'Intercept', 'Sun', et, 'lt+s', 'STEREO Ahead', spoint, alt
cspice_reclat, spoint, radius, lon, lat
lon = lon * (180/!dpi)
lat = lat * (180/!dpi)
```

The radius returned by CSPICE_RECLAT will be the solar radius of 696,000 km, and the sum radius+alt will be the heliocentric distance. Without aberration correction, the results will be identical with those from CSPICE_SPKEZR.

References

Fränz, M. and Harper, D. (2002). Heliospheric coordinate systems. *Planetary and Space Science*, **50**, 217–239.

- Hapgood, M. A. (1992). Space physics coordinate transformation: A user guide. *Planetary and Space Science*, **40**, 711–717.
- Russell, C. T. (1971). Geophysical coordinate transformations. Cosmic Electrodynamics, $\mathbf{2}$, 184–196.